

Laborator VBA pentru Excel 3

Vectori în VBA

Pentru a lucra cu vectori în Excel VBA, vom transmite ca parametrii în funcția VBA o **plajă de valori (Range)**.

Funcții Simple care returnează o singură valoare

Suma elementelor unui vector:

1. Lansați Microsoft Excel
2. Lansați Visual Basic Editor
3. In VBE adăugați un modul Module1
4. Tastați codul de mai jos in modulul Module1

Option Explicit

```
Function SumaVector(rng As Range) As Double
    Dim i As Integer
    Dim n As Integer
    Dim Suma As Double

    n = rng.Rows.Count
    ReDim Vect(1 To n) As Double

    Suma = 0
    For i = 1 To n
        Vect(i) = rng(i)
        Suma = Suma + Vect(i)
    Next i
    SumaVector = Suma
End Function
```

5. Lansați comanda **Compile VBAProject** din meniul **Debug** pentru a verifica codul
6. Activați foaia de calcul Excel și inserați următoarea serie de valori:

	A	B	C	D
1	1			
2	3			
3	5			
4	7			

5	9			
6	11			

- Mutați cursorul în celula B6 și lansați comanda **Insert Function**. Selectați funcția SumaVector în categoria **User Defined**.
- În fereastra **Function Arguments**, selectați plaja de celule A1:A6 și clic OK

Obs: Instrucțiunea **ReDim** permite dimensionarea sau redimensionarea unui tablou alocat dinamic care a fost declarat formal cu instrucțiunea **Dim**. Dacă instrucțiunea formală **Dim** lipsește, tabloul va fi declarat la prima dimensionare dinamică.

Range este un obiect Excel care poate fi o celulă, un rând, o coloană sau un grup de celule.

Rows este o proprietate a obiectului **Range**, iar **Count** este o proprietate a **Rows**. Proprietatea **Count** ne returnează numărul de rânduri ale obiectului Range.

Maximul elementelor unui vector:

- Adăugați funcția alăturată în modulul Module1.

```
Function MaxVector(rng As Range) As Double
    Dim i As Integer
    Dim n As Integer
    Dim Max As Double

    n = rng.Rows.Count
    ReDim Vect(1 To n) As Double
    For i = 1 To n
        Vect(i) = rng(i)
    Next i

    Max = Vect(1)
    For i = 2 To n
        If (Vect(i) > Max) Then
            Max = Vect(i)
        End If
    Next i
    MaxVector = Max
End Function
```

- Activați foaia de calcul Excel și inserați formula **MaxVector** în celula **C6**, selectând aceeași plajă de celule ca în exemplul precedent **=MaxVector(A1:A6)**.

Funcții simple care returnează un vector

Înmulțirea cu un scalar:

1. Adăugați funcția publică alăturată în modulul Module1
2. Activați foaia de calcul Excel și inserați formula **ScaleVector** în celula **D1**, selectând aceeași plajă de celule ca în exemplul precedent și dați valoarea 3 scalarului sc **=ScaleVector(A1:A6,3)**

```
Function ScaleVector(rng As Range, sc As Double) As Double()  
    Dim i As Integer  
    Dim n As Integer  
  
    ' Preluarea plajei de celule  
    n = rng.Rows.Count  
    ReDim Vect(1 To n) As Double  
    For i = 1 To n  
        Vect(i) = rng(i)  
    Next i  
  
    ReDim Scaled(1 To n) As Double  
    For i = 1 To n  
        Scaled(i) = sc * Vect(i)  
    Next i  
  
    ScaleVector = Scaled  
End Function
```

Obs: Remarcăm că la inserarea formulei **ScaleVector**, Excel va insera în celula care conține formula doar primul element al vectorului returnat de funcție. Pentru a insera întreg vectorul returnat de funcție, va trebui să folosim **Formulele matrice**.

Pentru a adăuga o formulă matrice procedați astfel:

1. Modificați prin editare formula din celula **D1** de forma **=TRANSPOSE(ScaleVector(A1:A6,3))**.
2. Selectați grupul de celule **D1:D6**.
3. Editați formula din **D1** apăsând **F2**.
4. Apăsăți **CTRL+SHIFT+ENTER**.

Pentru a găsi un singur element al vectorului returnat de funcție, putem folosi și funcția INDEX. De exemplu, pentru a insera într-o celulă elementul al patrulea al vectorului returnat de funcția **ScaleVector**, inserăm formula **=INDEX(ScaleVector(A1:A6,3),4)**.

Suma a 2 vectori:

1. Adăugați funcția publică alăturată în modulul Module1
2. Activați foaia de calcul Excel și inserați formula **Suma2Vec** în celula **E1**, și dați parametrii funcției plajele de celule **A1:A6** și **D1:D6**
3. Modificați formula din celula **E1**, imbricând-o în funcția **TRANSPOSE**, ca în observația precedentă
4. Adăugați o formulă matrice în plaja de celule **E1:E6**, cum a fost explicat în observația precedentă


```
Function Suma2Vec(rng1 As Range, rng2 As Range) As Variant
    Dim i As Integer
    Dim n1 As Integer
    Dim n2 As Integer

    ' Preluarea plajei de celule
    n1 = rng1.Rows.Count
    n2 = rng2.Rows.Count
    If (n1 <> n2) Then
        Suma2Vec = "Dimensiuni inegale."
        Exit Function
    End If
    ReDim Vect(1 To n1) As Double

    For i = 1 To n1
        Vect(i) = rng1(i) + rng2(i)
    Next i
    Suma2Vec = Vect
End Function
```

Sortarea unui vector

1. Adăugați funcția publică alăturată în modulul Module1.
2. Activați foaia de calcul Excel și adăugați seria de valori 1, 5, 4, 2, 3 în celulele **F1:F5**.
3. Inserați funcția **Bubble** în celula **G1**, și dați parametrii funcției plaja de celule **F1:F5**.
4. Modificați formula din celula **G1**, imbricând-o în funcția **TRANSPOSE**, ca în observația (Pentru a adăuga o formulă matrice).
5. Adăugați o formulă matrice în plaja de celule **G1:G5**, cum a fost explicat în observația (Pentru a adăuga o formulă matrice).
6. Adăugați un punct de oprire în linia **For i = 1 To n - 1**, și adăugați variabila de urmărit A la fereastra Inspector(**Watch**) cum a fost explicat în laboratorul 1 Funcții Simple la punctul Inspectarea variabilelor cu Inspectorul.
7. Executați codul pas cu pas și urmăriți modificarea vectorului A la fiecare pas de execuție.



Vom transcrie în VBA algoritmul de sortare Bubble Sort al cărui schemă logică este prezentată în figura alăturată.

La prima iterație după i , primele două elemente ale vectorului sunt comparate și sunt interschimbate dacă primul element este mai mare decât al doilea.

După aceea se compară al doilea element cu al treilea. Remarcați că în acest moment elementul al doilea poate fi de fapt primul element care a fost interschimbabil cu al doilea în comparația precedentă.

În final, în urma primei iterații după i elementul cel mai mare al vectorului este împins în ultima poziție.

În a doua iterație se continuă din

nou din prima poziție a vectorului până la penultima poziție, ș.a.m.d.

```

Public Function Bubble(rng As Range)
    Dim temp As Double
    Dim i As Integer, j As Integer, n As Integer

    n = rng.Rows.Count
    ReDim A(1 To n)
    For i = 1 To n
        A(i) = rng(i)
    Next i

    For i = 1 To n - 1
        For j = 1 To n - i
            If (A(j) > A(j + 1)) Then
                temp = A(j)
                A(j) = A(j + 1)
                A(j + 1) = temp
            End If
        Next j
    Next i
End Function
  
```

```
        A(j + 1) = temp
    End If
Next j
Next i
Bubble = A
End Function
```